HEWLETT PACKARD

Application Note 339-12

HP-IB Programming Hints for the HP 4194A

- HP 4194A Impedance/Gain-Phase Analyzer -



INTRODUCTION

The HP 4194A is an intelligent Impedance/ Gain-Phase Analyzer used for accurate evaluation of electronic components and circuits in the laboratory. In addition to the 4194A's uses in the lab, the HP 4194A's high measurement speed and internal Auto Sequence Program (ASP) will make a very effective contribution to high speed production line component testing.

This application note introduces programming techniques for building an HP 4194A HP-IB system controlled by an external computer. The techniques used to achieve high speed component testing and high speed data transfer to a computer are described. When used as a standalone instrument, through the use of ASP the HP 4194A has the capability to perform high speed measurements, parameter extractions, go/no-go testing and interfacing through an 8 bit I/O path to externally connected equipment. An HP-IB system comprised of an HP 4194A, an external computer, and using ASP can, in addition to the test and measurement function, store the extracted parameters without increasing the measurement cycle time. The HP 4194A can use the IEEE 64 bit binary data format to perform high speed data transfer to a computer.

In this application note programming hints for performing high speed parameter transfer are illustrated using a filter testing system as the example. In addition, an upload/download ASP program for backing up ASP programs to a mass storage device (floppy disc) is included.

MEASUREMENT DATA TRANSFER

Depending on your application, you can choose from three of the HP 4194A's data formats to obtain the most efficient data transfer to an external computer.

FMT1: ASCII mode
FMT2: Binary mode, IEEE 64 bit format
FMT3: Binary mode, IEEE 32 bit format

Format **FMT1** is most commonly used for general data transfer. Formats **FMT2** and **FMT3** are floating point binary formats specified in IEEE Standard 728-1982, and are useful for high speed data transfer. Table 1 lists typical data transfer times. The highest data transfer rate is obtained by using format **FMT3**, but format translation is required to put the data into a format usable by the computer after the data is transferred if the computer does not use the IEEE 32 bit format, such as the HP 9000 series 300 computers. Format **FMT2** is the data format used by HP Series 300 computers, so high speed data transfer and easy data manipulation are obtained without format translation. Program 1 demonstrates data transfer using data format **FMT1**, with the code in lines 140 and 150 performing the data transfer of 401 measurement points from the HP 4194A's **A** register to the computer.

Transferring data form the HP 4194A's internal registers to a computer can be performed in a manner similar to that shown in program 1. There are two types of registers in the 4194A which are described in Table 2. Type 1 registers are used for variable length data transfer like the **A** register, and the **NOP** command specifies the number of points to be transferred. If less than 401 points are to be transferred then use the Type 1 registers. Type 2 registers are used for fixed length data transfers, 401 points are always transferred.

Table 1 Data Transfer Speed (Typical)

Data Format	ENTER	TRANSFER
FMT1	740ms	------
FMT2	120ms	90ms
FMT3	50ms	40ms

* These are the times required to transfer 401 data points using an HP 9000 Series 300 Model 310 Computer.

```
10    ! RE-STORE "FMT1_ENT"
20    T=0
30    Tt=0
40    REAL A(0:400)
50    ASSIGN @Io TO 717
60    REMOTE @Io
70    OUTPUT @Io;"SWM2"
80    OUTPUT @Io;"FMT1"
90    FOR I=1 TO 10
100   OUTPUT @Io;"SWTRG"
110   WAIT 2.5
120   BEEP
130   T0=TIMEDATE
140   OUTPUT @Io;"A?"
150   ENTER @Io;A(*)
160   T1=TIMEDATE
170   BEEP
180   Tt=T1-T0
190   DISP Tt
200   T=T+Tt
210   NEXT I
220   PRINT DROUND(T/10*1000,6);"msec"
230   END
```

Program 1 Data Transfer Program using FMT1

```
10    ! RE-STORE "FMT2_TRF"
20    T=0
30    Tt=0
40    REAL A(0:400) BUFFER
50    DIM S$[4] BUFFER
60    ASSIGN @Ioo TO 717;FORMAT OFF
70    ASSIGN @Io TO 717;FORMAT ON
80    REMOTE @Io
90    OUTPUT @Io;"SWM2"
100   OUTPUT @Io;"FMT2"
110   FOR I=1 TO 10
120   OUTPUT @Io;"SWTRG"
130   WAIT 2.5
140   BEEP
150   ASSIGN @I_buff TO BUFFER A(*)
160   ASSIGN @Asc TO BUFFER S$
170   T0=TIMEDATE
180   OUTPUT @Io;"A?"
190   TRANSFER @Ioo TO @Asc;COUNT 4
200   TRANSFER @Ioo TO @I_buff;END,WAIT
210   T1=TIMEDATE
220   BEEP
230   Tt=T1-T0
240   DISP Tt
250   T=T+Tt
260   NEXT I
270   PRINT DROUND(T/10*1000,6);"msec"
280   END
```

Program 2 Data Transfer Program using FMT2

Table 2. Number of Data Points to Transfer

Array Registers	Number of Data Points to Transfer
A, B, SR, SX, OG, OB OFFSTA, OFFSTB, X	Number of points defined by the NOP command. Number of points defined in the Programmed Point Table when the Program Point Sweep is used.
C, D, E ~ J, RA ~ RL, TYG, TYB, MYG, MYB, TZR, TZX, MZR, MZX, TSTDR, TSTDX, MSTDR, MSTDX	401 Points

Sample Program 2 demonstrates high speed data transfer using data format **FMT2**. The **TRANSFER** command and **BUFFER** transfer is used to accomplish a higher data transfer rate than Program 1 which uses the **ENTER** command. Lines 150 and 160 assign an I/O path name to the buffer. At line 180, the data is ready to be output. Line 190 transfers 4 bits (the 4 bit header is stripped from the data) to the **S$** buffer. Line 200 need only to transfer 401 measurement point data only, no time is wasted transferring the header information. Program 2 can transfer 401 measurement point data in approximately 90 ms.

Sample Program 3 demonstrates high speed data transfer using data format **FMT3**. This program can transfer 401 measurement point data in approximately 40ms. However, after the data transfer is completed **7.3 SECONDS** are required for format conversion when using an HP 9000 series 300 computer.

```
10    ! RE-STORE "FMT3_TRF"
20    T10=0
30    T21=0
40    INTEGER A(1:802) BUFFER,Upper,Lower
50    REAL Aa(1:401)
60    DIM S$[4] BUFFER
70    ASSIGN @Ioo TO 717;FORMAT OFF
80    ASSIGN @Io TO 717;FORMAT ON
90    REMOTE @Io
100   OUTPUT @Io;"SWM2"
110   OUTPUT @Io;"FMT3"
120 FOR J=1 TO 10
130   OUTPUT @Io;"SWTRG"
140   WAIT 2.5
150   ASSIGN @I_buff TO BUFFER A(*)
160   ASSIGN @Asc TO BUFFER S$
170   T0=TIMEDATE
180   OUTPUT @Io;"A?"
190   TRANSFER @Ioo TO @Asc;COUNT 4
200   TRANSFER @Ioo TO @I_buff;END,WAIT
210   T1=TIMEDATE
220 Conversion:  !
230     FOR I=1 TO 401
240         Upper=A(I*2-1)
250         Lower=A(I*2)
260         IF Upper=0 AND Lower=0 THEN
270             Aa(I)=0
280             GOTO 380
290         END IF
300         Temp=Upper
310         Expo=SHIFT(SHIFT(Temp,-7),8)
320         Temp=SHIFT(SHIFT(Temp,-9),9)
330         Low=Lower
340         IF Lower<0 THEN Low=65536+Lower
350         Manti=Temp*2^16+Low
360         Aa(I)=DROUND(SGN(Upper)*(2^(Expo-127)+Manti*2^(Expo-150)),8)
370     NEXT I
380   T2=TIMEDATE
390   T10=T1-T0+T10
400   T21=T2-T1+T21
410 NEXT J
420   PRINT "TRF =",DROUND(T10/10*1000,6);"msec"
430   PRINT "CONV=",DROUND(T21/10*1000,6);"msec"
440   END
```

Program 3 Data Transfer Program using FMT3

HIGH SPEED FILTER TESTING

The following filter testing example is used to describe an efficient test system which includes device testing and data transfer on a production line environment, and for incoming/outgoing inspection.

Until now, it has been necessary to use an external computer to control measurement setup, measurement, data transfer, and to calculate the desired parameters from the measurement data. Instrument measurement speed can be increased but instrument control and data transfer will become the throughput bottle neck when you try to optimize throughput.

You can realize high speed device testing using ASP and an external computer. ASP is used to control the measurement process and parameter extraction. System throughput is optimized by transferring only the extracted parameters.

Figure 1 shows an example of an HP 4194A HP-IB system for testing band-pass filters, including measurement, sorting, automatic handler interfacing through an 8-bit I/O, and transferring the extracted parameters. Figure 2 is the timing chart for this system. After the device under test (DUT) is contacted and ready to test, the handler sends a **READY** signal to the HP 4194A through the 8-bit I/O interface, and the computer sends the ASP **RUN** command to the HP 4194A. On receiving the **RUN** command the HP 4194A executes the ASP program which performs measurement, parameter extraction, go/no-go comparison and handler interfacing through an 8-bit I/O. The handler sorts the test devices according to binning signals from the HP 4194A, and readies the next device for test. By using the 8-bit I/O and HP-IB bus, testing can be accomplished without timing errors between measurements, data transfer and device setup.



Figure 1 Band Pass Filter Testing System

The flowchart of a filter test program using a computer (A in Figure 1) with the HP 4194A is shown in Figure 3 and program 4 is a BASIC program based on this flowchart. The computer loads two ASP programs into the HP 4194A which initialzes the HP 4194A and performs the measurement. If the device to be tested requires a different instrument setup, you design, test, and debug an ASP program for the device, and then upload and store the ASP program on the computer's floppy disc. The ASP programs are developed in isolation from the computer program so debugging is simplified.



Figure 2 Example of Filter Testing Timing Chart

```
10    ! RE-STORE "AN339_12"
20    ! ###################################
30    !    FILTER TESTING SAMPLE PROGRAM
40    ! ###################################
50    OPTION BASE 1
60    Add=717                         ! ADDRESS FOR 4194A
70    N=10                            ! NUMBER OF MEASUREMENTS
80    T10=0
90    T21=0
100   REAL A(1:9) BUFFER
110   DIM Para(50,9)
120   DIM S$[4] BUFFER
130   ASSIGN @Ioo TO Add;FORMAT OFF
140   ASSIGN @Io TO Add;FORMAT ON
150   REMOTE @Io
160   OUTPUT @Io;"RST"
170   ! ###################################################
180   PRINTER IS 701
190   GOSUB Load_asp1
200   BEEP
210   Z=0
220   INPUT "OFFSET COMPENSATION ?   Y=1 / N='Continue'",Z
230   IF Z=1 THEN GOSUB Offset
240   GOSUB Load_asp2
250   GOSUB Testing
260   PRINTER IS CRT
270   STOP
280   ! ###################################################
290 Load_asp1:  !        LOAD ASP FOR INITIAL SETUP
300     Sp=SPOLL(@Io)
310     OUTPUT @Io;"LOAD980"
320     DISP "INITIAL SETTING !"
330     OUTPUT @Io;"RUN"
340     ON INTR 7 GOTO 370
350     ENABLE INTR 7;2
360     GOTO 360
370     DISABLE INTR 7
380     Sp=SPOLL(@Io)
390     IF BIT(Sp,5) THEN GOTO Error
400     RETURN
410 Offset:     !        OFFSET COMPENSATION
420     BEEP
430     DISP "MAKE THRU FOR OFFSET COMPENSATION !"
440     PAUSE
450     OUTPUT @Io;"RQS2;ITM2"
460     OUTPUT @Io;"SWTR6"
470     ON INTR 7 GOTO 500
480     ENABLE INTR 7;2
490     GOTO 490
500     Sp=SPOLL(@Io)
510     OUTPUT @Io;"OFSTR"
520     OUTPUT @Io;"RQS8;ITM;"
530     RETURN
540 Load_asp2:  !         LOAD ASP FOR TESTING
550     OUTPUT @Io;"LOAD981"
560     BEEP
570     DISP "CONNECT DEVICE"
580     PAUSE
590     RETURN
600 Testing:  !          TESTING
610   DISP "TESTING !"
620   FOR I=1 TO N
630   T0=TIMEDATE
640   OUTPUT @Io;"RUN"
650   ON INTR 7 GOTO Data_trf
660   ENABLE INTR 7;2
670   GOTO 670
```

Program 4 BASIC Program for Filter Testing



Figure 3 BASIC Program Flow Chart for Filter Testing

```
680 Data_trf:  !         DATA TRANSFER
690       T1=TIMEDATE
700       DISABLE INTR 7
710       Sp=SPOLL(@Ioo)
720       IF BIT(Sp,5) THEN GOSUB Error
730       ASSIGN @I_buff TO BUFFER A(*)
740       ASSIGN @Asc TO BUFFER S$
750       OUTPUT @Io;"SR?"
760       TRANSFER @Ioo TO @Asc;COUNT 4
770       TRANSFER @Ioo TO @I_buff;END,WAIT
780     FOR N=1 TO 9
790       Para(I,N)=A(N)
800     NEXT N
810   T2=TIMEDATE
820   T10=T10+(T1-T0)
830   T21=T21+(T2-T1)
840   NEXT I
850   DISP
860   PRINT "     TESTING  =";DROUND(T10/N*1000,6);"msec"
870   PRINT "DATA TRANSFER=";DROUND(T21/N*1000,6);"msec"
880   RETURN
890 Error:   !
900     BEEP
910     DISP "ERROR IN 4194A !      SPOLL IS",Sp
920     PAUSE
930     RETURN
940   END
```

Program 5 is an ASP initialization program for the HP 4194A. Store this program in the HP 4194A as file number 980.

Figure 4 is the flowchart for the measurement ASP program (B in Figure 3), and Program 6 is the measurement program based on this flow chart. Nine extracted parameters, such as insertion loss and ripple are stored in the array register **SR**. Thus, nine parameters can be transferred at once. Store this program in the HP 4194A as file number 981.

```
 ! ! SETUP FOR 21.4 MHZ FILTER TEST
10 RST
20 CMT"21.4 MHZ CRYSTAL B.P.F. TEST"
30 FNC2;GPP1;ATR2;NOP=201;OPB0;SWM2;AMAX=0;AMIN=-130
40 START=21.35MHZ;STOP=21.45MHZ
50 FMT2;RQS8
60 OPA0
70 ITM1
80 END
```

Program 5 ASP Program for Initial Setup

```
 10 ! 21.4 MHZ FILTER TEST
 20  NOP=201;AOF1;OUTPUT 00000000;INPUT R50
 30  IF R50=0 THEN GOTO 20
 40  SWTR6
 50  ANA0;MCF1;MKMXA;SR(1)=-MKRA;R9=MKR;DLCURS=-40;MCF4;R35=LCU
     RSL;R36=LCURSR;MCF0
 60  MCF1;MKR=R9;DLCURS=-65;MCF4;WIDTH;SR(5)=WID;MCF0
 70  MCF1;DLCURS=-6;MCF4;WIDTH;SR(3)=WID;SR(2)=SQR(LCURSR*LCURS
     L);MCF0
 80  MKR=R35;SMKR=START;MCF5;ARSTR;ANA1;LMN(A);MCF0
 90  IF SMKR<STOP  THEN GOTO 130
100  IF MKR>START  THEN GOTO 120
110  SR(6)=-R0-MKRA;SR(7)=MKR;GOTO 140
120  SMKR=START;MCF5;ARSTR;MKMXA;SR(6)=-R0-MKRA;SR(7)=MKR;MCF0;
     GOTO 140
130  MKR=START;MCF5;ARSTR;MKMXA;SR(6)=-R0-MKRA;SR(7)=MKR;MCF0
140  MKR=R36;SMKR=STOP;MCF5;ARSTR;LMN(A);MCF0
150  IF MKR>START THEN GOTO 170
160  SR(8)=-R0-SMKRA;SR(9)=SMKR;GOTO 180
170  SMKR=STOP;MCF5;ARSTR;MKMXA;SR(8)=-R0-MKRA;SR(9)=MKR;MCF0
180  MKR=LCURSL;SMKR=LCURSR;MCF5;ARSTR;LMX(A);MCF0
190  IF SMKR<STOP THEN GOTO 210
200  SR(4)=0;GOTO 220
210  MCF5;ARSTR;MKMNA;SR(4)=-R0-MKRA;MCF0
220  IF SR(1)>5 OR SR(4)>2 THEN GOTO 280
230  IF SR(2)<21.3995M OR SR(2)>21.4005M THEN GOTO 280
240  IF SR(3)<15K OR SR(5)>35K THEN GOTO 280
250  IF SR(6)<60 OR SR(8)<60 THEN GOTO 280
260  IF SR(7)>21.384M OR SR(9)<21.416M THEN GOTO 280
270  OUTPUT 00000001;NOP=9;GOTO 290
280  OUTPUT 00000010;NOP=9
290 END
```

Program 6 ASP Program for Testing



Figure 4 ASP Program Flowchart for Filter Testing

Figure 5 shows part of the data which is transferred. Using this sample program, it takes approximately 1.4s to sweep 201 measurement points, perform nine parameter calculations, perform comparison, and to output control signals to the handler. Approximately 27ms are required to transfer the parameters to the computer.



TESTING = 1414 msec
DATA TRANSFER= 27.005 msec

Parameters		NO.1	NO.2	NO.3	NO.4
IL	[dB]	3.96085	3.96042	3.96245	3.96187
CF	[Hz]	21.4000E+6	21.4000E+6	21.4000E+6	21.4000E+6
6dB BAND	[Hz]	17.3065E+3	17.3069E+3	17.3082E+3	17.3095E+3
Ripple	[dB]	4.82853	4.82107	4.82192	4.81882
65dB BAND	[Hz]	41.6261E+3	41.8809E+3	40.9333E+3	41.7023E+3
Spurious L	[dB]	65.0209	65.2221	64.8565	65.9157
Spurious L	[Hz]	21.3815E+6	21.3820E+6	21.3820E+6	21.3810E+6
Spurious R	[dB]	75.3262	75.3154	73.8715	74.1644
Spurious R	[Hz]	21.4485E+6	21.4500E+6	21.4175E+6	21.4175E+6

Figure 5 Filter Test Results

ASP UPLOAD/DOWNLOAD PROGRAM

The HP 4194A stores ASP programs in its 20K byte internal nonvolatile memory. Data can be stored for 2000 hours after the battery used in the nonvolatile memory circuit has been charged for 48 hours.

However, ASP programs should be backed up in the following cases:

- Many ASP programs for device testing are used.
- One HP 4194A is used by many people who have their own ASP programs.

```
10    ! 4194A ASP UPLOAD PROGRAM
20    DIM Prog$(300)[100]
30    DIM S$[20],Name$[10]
40    Add=717
50    !
60    BEEP
70    INPUT "Input ASP File number",No
80    OUTPUT Add;"LOAD";No
90    OUTPUT Add;"EDIT"
100   OUTPUT Add;"QUIT"
110   !
120   BEEP
130   INPUT "Enter File name to be stored",Name$
140   OUTPUT Add;"CPYM2"
150   OUTPUT Add;"COPY"
160   ENABLE INTR 7;2048
170   ON INTR 7 GOTO Store_disk
180   X=1
190   ENTER Add USING "%,20A";S$
200   ENTER Add;Prog$(X)
210   X=X+1
220   GOTO 200
230 Store_disk:!
240   DISABLE INTR 7
250   FOR I=1 TO X-1
260     Prog$(I)="PROG"&CHR$(39)&Prog$(I)&CHR$(39)
270     PRINT Prog$(I)
280   NEXT I
290   L=X-1
300   !
310   CREATE BDAT Name$,L+1,100
320   ASSIGN @File TO Name$
330   FOR I=1 TO L
340     OUTPUT @File,I;Prog$(I)
350   NEXT I
360   OUTPUT @File,L+1;"Eof"
370   BEEP
380   DISP " Up Load Complete"
390   END
```

Program 7 BASIC Program for ASP Upload

Programs 7 and 8 are useful utility programs for the above situations.

Program 7 is a BASIC program to upload and store ASP programs to the computers mass storage device, (floppy disc). Uploading is performed using the HP 4194A's **COPY** command, which prints out a listing of an ASP program and measurement data on a printer.

Program 8 downloads an ASP program from the computers mass storage device to the HP 4194A.

Use these BASIC programs to backup your library of ASP programs and to expand your uses for the HP 4194A.

```
10    ! 4194A ASP DOWNLOAD PROGRAM
20    DIM Prog$(300)[100]
30    DIM S$[20],Name$[10]
40    Add=717
50    Falcon=717
60    !
70    BEEP
80    INPUT "Input File Name to be downloaded",Name$
90    ASSIGN @File TO Name$
100   I=1
110   ENTER @File,I;Prog$(I)
120   PRINT Prog$(I)
130   IF Prog$(I)="Eof" THEN 160
140   I=I+1
150   GOTO 110
160   L=I-1
170   !
180   BEEP
190   DISP "Ready to SCRATCH, and press CONT"
200   PAUSE
210   OUTPUT Add;"SCRATCH"
220   FOR I=1 TO L
230     OUTPUT Add;Prog$(I)
240   NEXT I
250   OUTPUT Add;"QUIT"
260   BEEP
270   INPUT "Input ASP file number to be stored",Asp_no$
280   BEEP
290   INPUT "Input program comment",Comment$
300   OUTPUT Add;"STORE"&Asp_no$&",'"&Comment$&"'"
310   BEEP
320   DISP "Down Load Complete"
330   END
```

Program 8 BASIC Program for ASP Download

APPENDIX

Reference

Application Note 339-11
Filter Test for Production and Incoming Inspection using the HP 4194A

Instruments used in this Application Note

HP 4194A Impedance/Gain-Phase Analyzer
HP 9000 Series 300 Model 310 Computer
HP 9122D 3.5 inches Floppy Disc Drive
HP 2225A Think Jet Printer

For more information, call your local HP sales office listed in the telephone directory white pages. Ask for the Electronic Instrument Department, or write to Hewlett-Packard: **U.S.A.** - P.O. Box 10301, Palo Alto, CA 94303-0890. **Europe** - Hewlett-Packard S.A., P.O. Box 529, 1180 AM Amstelveen, The Netherlands. **Canada** - 6877 Goreway Drive, Mississauga, L4V 1M8, Ontario. **Japan** - Yokogawa-Hewlett-Packard Ltd., 3-29-21, Takaido-Higashi, Suginami-ku, Tokyo 168. **Far East** - Hewlett-Packard Asia Headquarters, 47/F China Resources Building, 26 Harbour Road, Wanchai Hong Kong. **Australasia** - Hewlett-Packard Australia Ltd., 31-41 Joseph Street, Blackburn, Victoria 3130 Australia. **Latin America** - Hewlett-Packard Latin America Headquarters, 3495 Deer Creek Rd., Palo Alto, CA 94304. For all other areas, please write to: Hewlett-Packard Intercontinental Headquarters, 3495 Deer Creek Rd., Palo Alto, CA 94304.

Printed in Japan **July, 1987** **5950-2929**